



SD Subsequent Function Requirement Routines

Technical Solution Guide

The Smith Consulting Group, Inc.
PO Box 703
Flanders, NJ 07836-0703
973-713-5846
SAPtechsolutions.com

SD Subsequent Function Requirement Routines

| | |
|--|----|
| Subsequent Function Requirement Routines | 3 |
| Subsequent Function Code | 3 |
| Table TVFO | 4 |
| Function Group V07A | 5 |
| Processing Function Modules | 6 |
| SD_ORDER_SUBSEQUENT_ALLOWED | 6 |
| SD_DELIVERY_SUBSEQUENT_ALLOWED | 7 |
| Multiple Calls..... | 8 |
| Custom Goods Issue Requirement Example..... | 9 |
| Business Requirement..... | 9 |
| Solution..... | 9 |
| Step 1 Clone the standard routine | 10 |
| Step 2 Activate the new routine | 13 |
| Step 3 Assign the new routine..... | 14 |
| Step 4 Code the routine..... | 16 |
| Step 5 Test the new routine..... | 20 |
| Step 6 Running the VOFM regeneration | 24 |

SD Subsequent Function Requirement Routines

Subsequent Function Requirement Routines

Subsequent Function Requirement Routines are used to control the subsequent processing of sales and delivery documents within the Sales and Distribution Module. The purpose of these routines is to prevent specific functions from occurring if certain conditions exist. These routines do not prevent subsequent documents from being created and they should not be confused with Copy Control. For example, these routines will not prevent an invoice from being created from a delivery, but they can prevent the picking of an existing delivery.

Subsequent Function Code

The system uses Subsequent Function Codes to determine which routines to call. Table TVFO contains all subsequent functions defined in the system. The following are the current functions.

| Function | Description |
|----------|--|
| 01 | Requirements from sales document (availability) |
| 02 | Purchase requisition from sales document |
| 03 | Requirements from sales document (procurement) |
| 04 | Create delivery due index |
| 05 | Production order/plan order from sales document |
| 06 | Release/Block-Assembly Order From Sales Document |
| 11 | Picking from delivery |
| 12 | Packing from delivery |
| 13 | Goods issue from delivery |

SD Subsequent Function Requirement Routines

Table TVFO

Table TVFO contains the routine mapping for each subsequent function code. For each function there are two routines specified. The System Routine (SyFORM routine) is the standard R/3 routine and should not be modified. The User Routine (Routine number) is the user defined requirement.

The screenshot shows the SAP Data Browser interface for Table TVFO. The window title is "Data Browser: Table TVFO Select Entries 10". The table has 5 columns: Client, Subseq. function, Group indicator, Routine number, and SyFORM routine. The data is as follows:


| Client | Subseq. function | Group indicator | Routine number | SyFORM routine |
|--------|------------------|-----------------|----------------|----------------|
| 400 | 01 | FOFU | 101 | 001 |
| 400 | 02 | FOFU | 102 | 002 |
| 400 | 03 | FOFU | 103 | 003 |
| 400 | 04 | FOFU | 104 | 004 |
| 400 | 05 | FOFU | 105 | 005 |
| 400 | 06 | FOFU | 106 | 000 |
| 400 | 11 | FOFU | 111 | 011 |
| 400 | 12 | FOFU | 112 | 012 |
| 400 | 13 | FOFU | 113 | 013 |
| 400 | 14 | FOFU | 114 | 014 |

In the example above, function code 11 is the picking function code. Routine 011 is the standard system routine and routine 111 is the user routine.

SD Subsequent Function Requirement Routines

Function Group V07A

Function group V07A contains the processing function modules and all of the requirement routines. Each routine is contained in a separate include. Both the include file and routine name contain the routine number.



The screenshot displays the SAP Function Builder interface for the function module `SD_DELIVERY_SUBSEQUENT_ALLOWED`. The left pane shows the object hierarchy for function group V07A, with the function module `SD_DELIVERY_SUBSEQUENT_ALLOWED` selected. The right pane shows the source code of the function module, which includes a local interface definition, a list of imported structures, and a series of conditional checks and data moves.

```
FUNCTION SD_DELIVERY_SUBSEQUENT_ALLOWED.
-----
***Lokale Schnittstelle:
IMPORTING
**  VALUE(BLIKP) LIKE LIKP STRUCTURE LIKP DEFAULT SPACE
**  VALUE(BLIPS) LIKE LIPSVB STRUCTURE LIPSVB
**          DEFAULT SPACE
**  VALUE(BTVLK) LIKE TVLK STRUCTURE TVLK DEFAULT SPACE
**  VALUE(BTVLP) LIKE TVLP STRUCTURE TVLP DEFAULT SPACE
**  VALUE(BVBUK) LIKE VBUKVB STRUCTURE VBUKVB
**          DEFAULT SPACE
**  VALUE(BVBUP) LIKE VBUPVB STRUCTURE VBUPVB
**          DEFAULT SPACE
**  VALUE(FOFUN) LIKE TVFO-FOFUN
**  VALUE(BVBPA) LIKE VBPA STRUCTURE VBPA DEFAULT SPACE
**  VALUE(BCALLER) OPTIONAL
EXCEPTIONS
**  ERROR
**  WARNING
-----
MOVE BLIKP TO LIKP.
MOVE BLIPS TO LIPS.
MOVE BVBUK TO VBUK.
MOVE BVBUP TO VBUP.
MOVE BTVLP TO TVLP.
MOVE BTVLK TO TVLK.
MOVE BCALLER TO CALLER.
* if blips-vbeln is initial.
* if blips-posnr eq '000000'.
IF BLIPS-POSNR EQ '000000' " Aufruf aus WA-Buchung
OR ( BVBPA-PARVW EQ 'WE' " Aufruf aus Kommiss.
AND NOT ( BVBPA-KUNNR IS INITIAL ) ).
MOVE BVBPA TO VBPA.
MOVE BVBPA TO EXP_CTR_VBPA.
ENDIF.
PERFORM TVFO_READ USING FOFUN.
* Systemprüfung
IF NOT TVFO-GRPNO_SY IS INITIAL.
MOVE TVFO-GRPNO_SY TO FORMNAME-GRPNO.
PERFORM (FORMNAME) IN PROGRAM (PROG_ORDER) .
ENDIF.
* Userprüfung
IF NOT TVFO-GRPNO IS INITIAL.
MOVE TVFO-GRPNO TO FORMNAME-GRPNO.
PERFORM (FORMNAME) IN PROGRAM (PROG_ORDER) .
ENDIF.
ENDFUNCTION.
```

SD Subsequent Function Requirement Routines

Processing Function Modules

There are two function modules in this function group that process requirements. One function is for sales documents, and the other is for delivery documents. These function modules are called from various points in the Order and Delivery programs. The calling programs pass the appropriate function code to the function module. The logic reads table TVFO with the passed function code to determine which routines to call. Each function executes the non-modifiable system routine first, and then executes the user defined routine.

SD_ORDER_SUBSEQUENT_ALLOWED

Passes all relevant sales document structures and the function code.

```
FUNCTION SD_ORDER_SUBSEQUENT_ALLOWED
IMPORTING
    BTVAP      LIKE  TVAP
    BTVEP      LIKE  TVEP
    BVBAK      LIKE  VBAK
    BVBAK      LIKE  VBAP
    BVBEK      LIKE  VBEP
    BVBEKPVB   LIKE  VBEPVB
    BVBUK      LIKE  VBUK
    BVBUP      LIKE  VBUP
    FOFUN      LIKE  TVFO-FOFUN
    I_SIMUL_MODE          DEFAULT SPACE
    I_CHECK_CYCLE_FOFUN_04  DEFAULT SPACE
EXPORTING
    NO_CREATE LIKE  TPTEST-STATUS
    NO_CHANGE LIKE  TPTEST-STATUS
EXCEPTIONS
    ERROR
    WARNING
    ERROR_BUT_DO_NOT_DELETE.
```

SD Subsequent Function Requirement Routines

SD_DELIVERY_SUBSEQUENT_ALLOWED

Passes all relevant delivery document structures and the function code.

FUNCTION SD_DELIVERY_SUBSEQUENT_ALLOWED

IMPORTING

| | | |
|---------|----------|------------|
| BLIKP | LIKE | LIKP |
| BLIPS | LIKE | LIPSVB |
| BTVLK | LIKE | TVLK |
| BTVLP | LIKE | TVLP |
| BVBUK | LIKE | VBUKVB |
| BVBUP | LIKE | VBUPVB |
| FOFUN | LIKE | TVFO-FOFUN |
| BVBPA | LIKE | VBPA |
| BCALLER | OPTIONAL | |


EXCEPTIONS

ERROR
WARNING.

SD Subsequent Function Requirement Routines

Multiple Calls

These functions are called multiple times from the calling program during the processing of a document. Each time the functions are called, the data contained in the structures may be different. In the case of delivery processing, the function is first called with only the header data populated. During subsequent calls, the header data is not passed, but item data is present.



The above example is the standard logic from the goods issue routine. The logic checks a field in VBUK only when LIPS is initial. The ELSE side of the IF statement is for checking item fields. This is because when LIPS is initial, the header data is being passed.

We suggest setting break points in the functions and routines to understand exactly how they work before implementing your own logic.

SD Subsequent Function Requirement Routines

Custom Goods Issue Requirement Example

The following example demonstrates how to implement a custom goods issue requirement routine. The basic process is the same for the other available functions.

Business Requirement

For a specific plant, inventory balances are maintained in a custom 'Z' table in addition to the standard R/3 inventory. This was done as part of a larger project to track the quantities of materials imported into Mexico for re-sale. Without elaborating on the design of the entire project, the purpose of this portion is to ensure that inventory exists in the 'Z' table during the goods issue process. If there is insufficient inventory in the table, the goods issue will not be posted.


Solution

To solve this problem, a custom goods issue requirement routine will be implemented. The logic in this routine will check the inventory balance in the 'Z' table and prevent the goods issue if the quantity in the table is less than the delivery quantity.

SD Subsequent Function Requirement Routines

Step 1 *Clone the standard routine*

Using transaction VOFM, navigate to the subsequent function requirement routine for goods issue. Alternatively, you can go directly to the good issue requirement using transaction OVB7.



SD Subsequent Function Requirement Routines


There is only one routine allowed for each function even though the screen implies that multiple routines can be used. On this screen, press the 'Req. maintenance' button at the bottom of the screen.

The screenshot shows the SAP interface for the 'Change View Requirement for Goods Issue from a Delivery' screen. The title bar indicates 'Table view', 'Edit', 'Goto', 'Selection criteria', 'Utilities', 'System', and 'Help'. The main area displays a table titled 'Goods issue for delivery' with the following columns: 'Routin...', 'Requirmnt', 'SyFORM rou...', 'System requir.', and 'Subs.fu...'. The table contains one row with the following values: 113, Goods Issue, 13, System: Goods Issue, and 13. Below the table is a 'Req. maintenance' button and a status bar indicating 'Only one entry chosen'.

| Routin... | Requirmnt | SyFORM rou... | System requir. | Subs.fu... |
|-----------|-------------|---------------|---------------------|------------|
| 113 | Goods Issue | 13 | System: Goods Issue | 13 |

SD Subsequent Function Requirement Routines

To clone the standard routine (113), type over the existing routine number and press enter. The system will prompt you for an object key. Obtain and specify the object key, then press ENTER.



The screenshot shows the SAP SD Subsequent Function Requirement Routines table. The table has four columns: Routine number, Description, Active, and Application. The routine number 913 is highlighted in yellow, and its description is 'Custom Goods Issue'. The other routines listed are standard SAP routines with descriptions like 'System: BANF', 'System: Requirements', etc.


| Routine number | Description | Active | Application |
|----------------|----------------------|-------------------------------------|-------------|
| 2 | System: BANF | <input checked="" type="checkbox"/> | |
| 3 | System: Requirements | <input checked="" type="checkbox"/> | |
| 4 | System: DlvDueIndex | <input checked="" type="checkbox"/> | |
| 5 | System: Assembly | <input checked="" type="checkbox"/> | |
| 11 | System: Picking | <input checked="" type="checkbox"/> | |
| 12 | System: Packing | <input checked="" type="checkbox"/> | |
| 13 | System: Goods Issue | <input checked="" type="checkbox"/> | |
| 14 | System: Distribution | <input checked="" type="checkbox"/> | |
| 101 | Reqs.availability | <input checked="" type="checkbox"/> | |
| 102 | PReq | <input checked="" type="checkbox"/> | |
| 103 | Requirements | <input checked="" type="checkbox"/> | |
| 104 | Delivery Due Index | <input checked="" type="checkbox"/> | |
| 105 | Assembly | <input checked="" type="checkbox"/> | |
| 106 | Block construction | <input checked="" type="checkbox"/> | |
| 108 | Ship. due date index | <input checked="" type="checkbox"/> | |
| 111 | Picking | <input checked="" type="checkbox"/> | |
| 112 | Packing | <input checked="" type="checkbox"/> | |
| 913 | Custom Goods Issue | <input checked="" type="checkbox"/> | |
| 114 | Distribution | <input checked="" type="checkbox"/> | |

In this example, we typed routine 913 over 113. It is a common practice to simply replace the first digit of these routines with a '9'. Using this standard makes it obvious which routine was used for cloning.

SD Subsequent Function Requirement Routines

Step 2 *Activate the new routine*

VOFM routines need to be activated before they can be used.



SD Subsequent Function Requirement Routines

Examining the TVFO table, we now see that subsequent function 13 points to routine number 913.

Data Browser: Table TVFO Select Entries 10


Table : TVFO
Displayed fields: 5 of 5 Fixed columns: 2 List width 0250

| Client | Subseq.function | Group indicator | Routine number | SyFORM routine |
|------------------------------|-----------------|-----------------|----------------|----------------|
| <input type="checkbox"/> 400 | 01 | FOFU | 101 | 001 |
| <input type="checkbox"/> 400 | 02 | FOFU | 102 | 002 |
| <input type="checkbox"/> 400 | 03 | FOFU | 103 | 003 |
| <input type="checkbox"/> 400 | 04 | FOFU | 104 | 004 |
| <input type="checkbox"/> 400 | 05 | FOFU | 105 | 005 |
| <input type="checkbox"/> 400 | 06 | FOFU | 106 | 000 |
| <input type="checkbox"/> 400 | 11 | FOFU | 111 | 011 |
| <input type="checkbox"/> 400 | 12 | FOFU | 112 | 012 |
| <input type="checkbox"/> 400 | 13 | FOFU | 913 | 013 |
| <input type="checkbox"/> 400 | 14 | FOFU | 114 | 014 |

SD Subsequent Function Requirement Routines

Step 4 Code the routine

To access the source code, pull down the list of valid routines (F4), select the new routine and press the source code icon at the bottom of the dialog box.



SD Subsequent Function Requirement Routines

When a routine is cloned, the system automatically inserts comments at the top of the include denoting that it was copied from another routine. Since the include generated (RV07A913) is within the R/3 name range, the system protects the code as if we were installing a source code modification.


```
*** Attention: copied routine! ***
*** Attention: ***
*** Character string 113 is replaced by 913 everywhere ! ***
*****

*-----*
* FORM BEDINGUNG_PRUEFEN_913 *
* User checks for subsequent functions from a delivery *
* Goods issue *
*-----*

FORM BEDINGUNG_PRUEFEN_913.
* Goods issue is not allowed if a credit block exists
IF LIPS EQ SPACE.
* document header
IF VBUK-CMGST CA 'B'.
MESSAGE ID 'V1' TYPE 'E' NUMBER '183' RAISING ERROR.
ENDIF.
ELSE.
* line item
ENDIF.
ENDFORM.
*eject
```


SD Subsequent Function Requirement Routines

For custom routines, it is acceptable to turn off the modification assistant as long as the routine number begins with a '9'.



SD Subsequent Function Requirement Routines

Our custom logic is added to the routine. The standard logic that was already present is kept.



```

*-----*
* Goods issue *
*-----*
FORM bedingung_pruefen_913.
  TABLES: zzped_mat_link.
  DATA:   v_qty          LIKE lips-lfimg.

  CLEAR v_qty.
  * Goods issue is not allowed if a credit block exists
  IF lips EQ space.
  * document header
  IF vbuk-cmgst CA 'B'.
    MESSAGE ID 'V1' TYPE 'E' NUMBER '183' RAISING error.
  ENDIF.
  ELSE.
  * line item
  CHECK lips-werks = '1500'.
  CHECK lips-lfimg GT 0.
  * The following check is for Mexican products only, the ones that
  * have pedimento number assigned to them.
  SELECT
    zz_pedng
    INTO   zzped_mat_link-zz_pedng
    FROM   zzped_mat_link
    WHERE  matnr = lips-matnr.


    v_qty          = v_qty + zzped_mat_link-zz_pedng.
  ENDSELECT.
  IF sy-subrc = 0.
    IF ( lips-bwart = '601' OR
        lips-bwart = '658' ).
      IF lips-lfimg > v_qty.
        MESSAGE ID 'ZV' TYPE 'E' NUMBER '058' WITH lips-matnr
          RAISING error.
      ENDIF.
    ENDIF.
  ENDIF.
ENDFORM.

```

SD Subsequent Function Requirement Routines


Step 5 **Test the new routine**

We have selected a delivery and pressed the Post goods issue button.



SD Subsequent Function Requirement Routines


A break point was set in the SD_DELIVERY_SUBSEQUENT_ALLOWED function module.



We can see that the function was called with a function code of '13'. The logic read table TVFO and determined the system and user defined routine numbers. The logic will build the form names with these numbers and execute the system form first and then the user form.

SD Subsequent Function Requirement Routines

The program is now in our custom routine. The inventory in our 'Z' table is not enough to cover the goods issue. The logic will issue an error using the MESSAGE statement. This will be trapped and the message will be displayed on the error log.



The screenshot displays the SAP ABAP Debugger interface. The main window shows the source code of a custom routine named 'FORM BEDINGUNG_PRUEFEN_913'. The code includes a SELECT statement to retrieve data from 'zzped_mat_link' and a logic block that checks for a specific error condition. A red 'STOP' icon indicates that the debugger has paused execution at the MESSAGE statement.

```
FORM BEDINGUNG_PRUEFEN_913
  SELECT      zz_pedng
             INTO  zzped_mat_link-zz_pedng
             FROM  zzped_mat_link
             WHERE matnr = lips-matnr.

  v_qty      = v_qty + zzped_mat_link-zz_pedng.
ENDSELECT.
IF sy-subrc = 0.
  IF ( lips-bwart = '601' OR
      lips-bwart = '658' ).
    IF lips-lfimg > v_qty.
      MESSAGE ID 'ZV' TYPE 'E' NUMBER '058' WITH lips-matnr
              RAISING error.
    ENDIF.
  ENDIF.
ENDIF.
ENDIF.
ENDFORM.
```


Below the code editor, the 'Field names' table shows the current values of variables:

| Field names | Field contents |
|-------------|----------------|
| sy-subrc | 0 |
| lips-lfimg | 12.000 |
| v_qty | 0.000 |

At the bottom of the debugger, the status bar shows: SY-SUBRC 0, SY-TABIX 3, and SY-DBCNT 0. A message at the very bottom states 'Field copied into the display'.

SD Subsequent Function Requirement Routines


The error generated in the routine displays on the goods movement log. The goods issue did not post.



SD Subsequent Function Requirement Routines

Step 6 *Running the VOFM regeneration*

Whenever a VOFM routine is created, a special generation program (RV80HGEN) should be run in each system that the routine is moved to.



Note: This is a critical step that should not be overlooked. If this step is not executed, short dumps can result.

SD Subsequent Function Requirement Routines

The RV80HGEN completed.

